

Chapter 13

The LOAN Procedure

Chapter Table of Contents

OVERVIEW	643
GETTING STARTED	644
Analyzing Fixed Rate Loans	644
Analyzing Balloon Payment Loans	646
Analyzing Adjustable Rate Loans	647
Analyzing Buydown Rate Loans	648
Loan Repayment Schedule	648
Loan Comparison	650
SYNTAX	653
Functional Summary	653
PROC LOAN Statement	655
FIXED Statement	655
BALLOON Statement	659
ARM Statement	660
BUYDOWN Statement	662
COMPARE Statement	663
DETAILS	665
Computational Details	665
Loan Comparison Details	666
OUT= Data Set	668
OUTCOMP= Data Set	668
OUTSUM= Data Set	669
Printed Output	670
ODS Table Names	671
EXAMPLES	673
Example 13.1 Discount Points for Lower Interest Rates	673
Example 13.2 Refinancing a Loan	674
Example 13.3 Prepayments on a Loan	675
Example 13.4 Output Data Sets	677
REFERENCES	679

Chapter 13

The LOAN Procedure

Overview

The LOAN procedure analyzes and compares fixed rate, adjustable rate, buydown, and balloon payment loans. The LOAN procedure computes the loan parameters and outputs the loan summary information for each loan.

Multiple loans can be processed and compared in terms of economic criteria such as after-tax or before-tax present worth of cost and true interest rate, breakeven of periodic payment and of interest paid, and outstanding balance at different periods in time. PROC LOAN selects the best alternative in terms of the specified economic criterion for each loan comparison period.

The LOAN procedure allows various payment and compounding intervals (including continuous compounding) and uniform or lump sum prepayments for each loan. Down payments, discount points, and other initialization costs can be included in the loan analysis and comparison.

Getting Started

PROC LOAN supports four types of loans. You specify each type of loan using the corresponding statement: FIXED, BALLOON, ARM, and BUYDOWN.

- **FIXED** - Fixed rate loans have a constant interest rate and periodic payment throughout the life of the loan.
- **BALLOON** - Balloon payment loans are fixed rate loans with lump sum payments in certain payment periods in addition to the constant periodic payment.
- **ARM** - Adjustable rate loans are those in which the interest rate and periodic payment vary over the life of the loan. The future interest rates of an adjustable rate loan are not known with certainty, but they will vary within specified limits according to terms stated in the loan agreement. In practice, the rate adjustment terms vary. PROC LOAN offers a flexible set of options to capture a wide variety of rate adjustment terms.
- **BUYDOWN** - Buydown rate loans are similar to adjustable rate loans, but the interest rate adjustments are predetermined at the initialization of the loan, usually by paying interest points at the time of loan initialization.

Analyzing Fixed Rate Loans

The most common loan analysis is the calculation of the periodic payment when the loan amount, life, and interest rate are known. The following PROC LOAN statements analyze a 15-year (180 monthly payments) fixed rate loan for \$100,000 with an annual nominal interest rate of 13%:

```
proc loan;  
    fixed amount=100000 rate=13 life=180;  
run;
```

Another parameter PROC LOAN can compute is the maximum amount you can borrow given the periodic payment you can afford and the rates available in the market. The following SAS statements analyze a loan for 180 monthly payments of \$1250, with a nominal annual rate of 13%:

```
proc loan;  
    fixed payment=1250 rate=13 life=180;  
run;
```

Assume that you want to borrow \$100,000 and can pay \$1250 a month. You know that the lender charges a 13% nominal interest rate compounded monthly. To determine how long it will take you to pay off your debt, use the following statements:

```
proc loan;  
    fixed amount=100000 payment=1250 rate=13;  
run;
```

Sometimes, a loan is expressed in terms of the amount borrowed and the amount and number of periodic payments. In this case, you want to calculate the annual nominal rate charged on the loan to compare it to other alternatives. The following statements analyze a loan of \$100,000 paid in 180 monthly payments of \$1250:

```
proc loan;
    fixed amount=100000 payment=1250 life=180;
run;
```

There are four basic parameters that define a loan: life (number of periodic payments), principal amount, interest rate, and the periodic payment amount. PROC LOAN calculates the missing parameter among these four. Loan analysis output includes a loan summary table and an amortization schedule.

You can use the START= and LABEL= options to enhance your output. The START= option specifies the date of loan initialization and dates all the output accordingly. The LABEL= specification is used to label all output corresponding to a particular loan and is especially useful when multiple loans are analyzed. For example, the preceding statements for the first fixed rate loan are revised to include the START= and LABEL= options as follows:

```
proc loan start=1992:12;
    fixed amount=100000 rate=13 life=180
        label='BANK1, Fixed Rate';
run;
```

Loan Summary Table

The loan summary table is produced by default and contains loan analysis information. It shows the principal amount, the costs at the time of loan initialization (downpayment, discount points, and other loan initialization costs), the total payment and interest, the initial nominal and effective interest rates, payment and compounding intervals, the length of the loan in the time units specified, the start and end dates (if specified), a list of nominal and effective interest rates, and periodic payments throughout the life of the loan.

Figure 13.1 shows the loan summary table for the fixed rate loan labeled "BANK1, Fixed Rate."

The LOAN Procedure			
Fixed Rate Loan Summary			
BANK1, Fixed Rate			
Downpayment	0.00	Principal Amount	100000.00
Initialization	0.00	Points	0.00
Total Interest	127744.33	Nominal Rate	13.00%
Total Payment	227744.33	Effective Rate	13.80%
Pay Interval	MONTHLY	Compounding	MONTHLY
No. of Payments	180	No. of Compoundings	180
Start Date	DEC1992	End Date	DEC2007

Rates and Payments for BANK1, Fixed Rate			
Date	Nominal Rate	Effective Rate	Payment
DEC1992	13.00%	13.80%	1265.24

Figure 13.1. Fixed Rate Loan Summary

The loan is initialized in December, 1992 and paid off in December, 2007. The monthly payment is calculated to be \$1265.24, and the effective interest rate is 13.8%. Over the 15 years, \$127,744.33 is paid for interest charges on the loan.

Analyzing Balloon Payment Loans

You specify balloon payment loans like fixed rate loans, with the additional specification of the balloon payments. Assume you have an alternative to finance the \$100,000 investment with a 15-year balloon payment loan. The annual nominal rate is 13%, as in the fixed rate loan. The terms of the loan require two balloon payments of \$2000 and \$1000 at the 15th and 48th payment periods, respectively. These balloon payments keep the periodic payment lower than that of the fixed rate loan. The balloon payment loan is defined by the following BALLOON statement:

```
proc loan start=1992:12;
    balloon amount=100000 rate=13 life=180
        balloonpayment=(15=2000 48=1000)
        label='BANK2, with Balloon Payment';
run;
```

List of Balloon Payments

In addition to the information for the fixed rate loan, the "Loan Summary Table" for the balloon payment loan includes a list of balloon payments in the "List of Rates and Payments." For example, the balloon payment loan described previously includes two balloon payments, as shown in Figure 13.2.

The LOAN Procedure			
Rates and Payments for BANK2, with Balloon Payment			
Date	Nominal Rate	Effective Rate	Payment
DEC1992	13.00%	13.80%	1236.17
	Balloon Period	Payment	
	MAR1994	2000.00	
	DEC1996	1000.00	

Figure 13.2. List of Rates and Payments for a Balloon Payment Loan

The periodic payment for the balloon payment loan is \$29.07 less than that of the fixed rate loan.

Analyzing Adjustable Rate Loans

In addition to specifying the basic loan parameters, you need to specify the terms of the rate adjustments for an adjustable rate loan. There are many ways of stating the rate adjustment terms, and PROC LOAN facilitates all of them. For details, see the "Rate Adjustment Terms Options" in the "ARM Statement" section later in this chapter.

Assume that you have an alternative to finance the \$100,000 investment with a 15-year adjustable rate loan with an initial annual nominal interest rate of 11%. The rate adjustment terms specify a 0.5% annual cap, a 2.5% life cap, and a rate adjustment every 12 months. *Annual cap* refers to the maximum increase in interest rate per adjustment period, and *life cap* refers to the maximum increase over the life of the loan. The following ARM statement specifies this adjustable rate loan assuming the interest rate adjustments will always increase by the maximum allowed by the terms of the loan. These assumptions are specified by the WORSTCASE and CAPS= options, as shown in the following statements:

```
proc loan start=1992:12;
  arm amount=100000 rate=11 life=180 worstcase
    caps=(0.5, 2.5)
    label='BANK3, Adjustable Rate';
run;
```

List of Rates and Payments for Adjustable Rate Loans

The "List of Rates and Payments" in the loan summary table for the adjustable rate loans reflects the changes in the interest rates and payments, as well as the dates these changes become effective. For the adjustable rate loan described previously, Figure 13.3 shows the "List of Rates and Payments" indicating five annual rate adjustments in addition to the initial rate and payment.

The LOAN Procedure			
Rates and Payments for BANK3, Adjustable Rate			
Date	Nominal Rate	Effective Rate	Payment
DEC1992	11.00%	11.57%	1136.60
JAN1994	11.50%	12.13%	1166.74
JAN1995	12.00%	12.68%	1195.72
JAN1996	12.50%	13.24%	1223.43
JAN1997	13.00%	13.80%	1249.75
JAN1998	13.50%	14.37%	1274.55

Figure 13.3. List of Rates and Payments for an Adjustable Rate Loan

Notice that the periodic payment of the adjustable rate loan as of January 1998 (\$1274.55) exceeds that of the fixed rate loan (\$1265.24).

Analyzing Buydown Rate Loans

A 15-year buydown rate loan is another alternative to finance the \$100,000 investment. The nominal annual interest rate is 12% initially and will increase to 14% and 16% as of the 24th and 48th payment periods, respectively. The nominal annual interest rate is lower than that of the fixed rate alternative, at the cost of a 1% discount point (\$1000) paid at the initialization of the loan. The following BUYDOWN statement represents this loan alternative:

```
proc loan start=1992:12;
  buydown amount=100000 rate=12 life=180
    buydownrates=(24=14 48=16) pointpct=1
    label='BANK4, Buydown';
run;
```

List of Rates and Payments for Buydown Rate Loans

Figure 13.4 shows the "List of Rates and Payments" in the loan summary table. It reflects the two rate adjustments and the corresponding monthly payments as well as the initial values for these parameters. As of December 1994, the periodic payment of the buydown loan exceeds the periodic payment for any of the other alternatives.

The LOAN Procedure			
Rates and Payments for BANK4, Buydown			
Date	Nominal Rate	Effective Rate	Payment
DEC1992	12.00%	12.68%	1200.17
DEC1994	14.00%	14.93%	1320.31
DEC1996	16.00%	17.23%	1432.33

Figure 13.4. List of Rates and Payments for a Buydown Rate Loan

Loan Repayment Schedule

In addition to the loan summary, you can print a loan repayment (amortization) schedule for each loan. For each payment period, this schedule contains the year and period within the year (or date, if the START= option is specified), the principal balance at

the beginning of the period, the total payment, interest payment, principal repayment for the period, and the principal balance at the end of the period.

To print the first year of the amortization schedule for the fixed rate loan shown in Figure 13.5, use the following statements:

```
proc loan start=1992:12;
    fixed amount=100000 rate=13 life=180
        label='BANK1, Fixed Rate'
        schedule=1;
run;
```

The LOAN Procedure					
Loan Repayment Schedule					
BANK1, Fixed Rate					
Date	Beginning Outstanding	Payment	Interest Payment	Principal Repayment	Ending Outstanding
DEC1992	100000.00	0.00	0.00	0.00	100000.00
DEC1992	100000.00	0.00	0.00	0.00	100000.00
JAN1993	100000.00	1265.24	1083.33	181.91	99818.09
FEB1993	99818.09	1265.24	1081.36	183.88	99634.21
MAR1993	99634.21	1265.24	1079.37	185.87	99448.34
APR1993	99448.34	1265.24	1077.36	187.88	99260.46
MAY1993	99260.46	1265.24	1075.32	189.92	99070.54
JUN1993	99070.54	1265.24	1073.26	191.98	98878.56
JUL1993	98878.56	1265.24	1071.18	194.06	98684.50
AUG1993	98684.50	1265.24	1069.08	196.16	98488.34
SEP1993	98488.34	1265.24	1066.96	198.28	98290.06
OCT1993	98290.06	1265.24	1064.81	200.43	98089.63
NOV1993	98089.63	1265.24	1062.64	202.60	97887.03
DEC1993	97887.03	1265.24	1060.44	204.80	97682.23
DEC1993	100000.00	15182.88	12865.11	2317.77	97682.23

Figure 13.5. Loan Repayment Schedule for the First Year

The principal balance at the end of one year is \$97,682.23. The total payment for the year is \$15,182.88 of which \$2317.77 went towards principal repayment.

You can also print the amortization schedule with annual summary information or for a specified number of years. The SCHEDULE=YEARLY option produces an annual summary loan amortization schedule, which is useful for loans with long life. For example, to print the annual summary loan repayment schedule for the buydown loan shown in Figure 13.6, use the following statements:

```
proc loan start=1992:12;
    buydown amount=100000 rate=12 life=180
        buydownrates=(24=14 48=16) pointpct=1
        schedule=yearly
        label='BANK4, Buydown';
run;
```

The LOAN Procedure					
Loan Repayment Schedule					
BANK4, Buydown					
Year	Beginning Outstanding	Payment	Interest Payment	Principal Repayment	Ending Outstanding
1992	100000.00	1000.00	0.00	0.00	100000.00
1993	100000.00	14402.04	11863.39	2538.65	97461.35
1994	97461.35	14522.18	11699.51	2822.67	94638.68
1995	94638.68	15843.72	13076.31	2767.41	91871.27
1996	91871.27	15955.74	12811.27	3144.47	88726.80
1997	88726.80	17187.96	13966.84	3221.12	85505.68
1998	85505.68	17187.96	13411.96	3776.00	81729.68
1999	81729.68	17187.96	12761.44	4426.52	77303.16
2000	77303.16	17187.96	11998.90	5189.06	72114.10
2001	72114.10	17187.96	11104.96	6083.00	66031.10
2002	66031.10	17187.96	10057.05	7130.91	58900.19
2003	58900.19	17187.96	8828.61	8359.35	50540.84
2004	50540.84	17187.96	7388.53	9799.43	40741.41
2005	40741.41	17187.96	5700.35	11487.61	29253.80
2006	29253.80	17187.96	3721.38	13466.58	15787.22
2007	15787.22	17188.72	1401.50	15787.22	0.00

Figure 13.6. Annual Summary Loan Repayment Schedule

Loan Comparison

The LOAN procedure can compare alternative loans on the basis of different economic criteria and help select the most desirable loan. You can compare alternative loans through different points in time. The economic criteria offered by PROC LOAN are

- outstanding principal balance, that is, the unpaid balance of the loan
- present worth of cost, that is, before-tax or after-tax net value of the loan cash flow through the comparison period. The cash flow includes all payments, discount points, initialization costs, down payment, and the outstanding principal balance at the comparison period.
- true interest rate, that is, before-tax or after-tax effective annual interest rate charged on the loan. The cash flow includes all payments, discount points, initialization costs, and the outstanding principal balance at the specified comparison period.
- periodic payment
- the total interest paid on the loan

The figures for present worth of cost, true interest rate, and interest paid are reported on the cash flow through the comparison period. The reported outstanding principal balance and the periodic payment are the values as of the comparison period.

The COMPARE statement specifies the type of comparison and the periods of comparison. For each period specified in the compare statement, a loan comparison report is printed that also indicates the best alternative. Different criteria may lead to selection of different alternatives. Also, the period of comparison may change the desirable alternative. See the section "Loan Comparison Details" later in this chapter for further information.

Comparison of 15-Year versus 30-Year Loan Alternatives

An issue that arises in the purchase of a house is the length of the loan life. In the U.S., residential home loans are usually for 15 or 30 years. Ordinarily, 15-year loans have a lower interest rate but higher periodic payments than 30-year loans. A comparison of both loans might identify the better loan for your means and needs. The following SAS statements compare two such loans:

```
proc loan start=1992:12 amount=100000;
  fixed rate=13 life=360 label='30 year loan';
  fixed rate=12.5 life=180 label='15 year loan';
  compare;
run;
```

Default Loan Comparison Report

The default loan comparison report in Figure 13.7 shows the ending outstanding balance, periodic payment, interest paid, and before-tax true rate at the end of 30 years. In the case of the default loan comparison, the selection of the best alternative is based on minimization of the true rate.

The LOAN Procedure				
Loan Comparison Report				
Analysis through DEC2022				
Loan Label	Ending Outstanding	Payment	Interest Paid	True Rate
30 year loan	0.00	1105.57	298231.4	13.80
15 year loan	0.00	1233.78	121854.9	13.24

NOTE: "15 year loan" is the best alternative based on true rate analysis through DEC2022.

Figure 13.7. Default Loan Comparison Report

Based on true rate, the best alternative is the 15-year loan. However, if the objective were to minimize the periodic payment, the 30-year loan would be the more desirable.

Comparison of Fixed Rate and Adjustable Rate Loans

Suppose you want to compare a fixed rate loan to an adjustable rate alternative. The nominal interest rate on the adjustable rate loan is initially 1.5% lower than the fixed rate loan. The future rates of the adjustable rate loan are calculated using the worst case scenario.

According to current U.S. tax laws, the loan for a family home qualifies the interest paid on the loan as a tax deduction. The TAXRATE=33 (income tax rate) option on the compare statement bases the calculations of true interest rate on the after-tax cash flow. Assume, also, that you are uncertain as to how long you will keep this property. The AT=(60 120) option, as shown in the following example, produces two loan comparison reports through the end of the 5th and the 10th years, respectively:

```
proc loan start=1992:12 amount=100000 life=360;
  fixed rate=13 label='BANK1, Fixed Rate' ;
  arm rate=11.5 worstcase caps=(0.5, 2.5)
  label='BANK3, Adjustable Rate';
```

Part 2. General Information

```
compare taxrate=33 at=(60 120);  
  
run;
```

After-Tax Loan Comparison Reports

The two loan comparison reports in Figure 13.8 and Figure 13.9 show the ending outstanding balance, periodic payment, interest paid, and after-tax true rate at the end of five years and ten years, respectively.

The LOAN Procedure				
Loan Comparison Report				
Analysis through DEC1997				
Loan Label	Ending Outstanding	Payment	Interest Paid	True Rate
BANK1, Fixed Rate	98081.66	1106.20	64453.66	9.07
BANK3, Adjustable Rate	97897.43	1141.14	61848.19	8.64

NOTE: "BANK3, Adjustable Rate" is the best alternative based on true rate analysis through DEC1997.

Figure 13.8. Loan Comparison Report as of December 1997

The LOAN Procedure				
Loan Comparison Report				
Analysis through DEC2002				
Loan Label	Ending Outstanding	Payment	Interest Paid	True Rate
BANK1, Fixed Rate	94419.82	1106.20	127163.8	9.07
BANK3, Adjustable Rate	94767.36	1178.45	129425.1	9.08

NOTE: "BANK1, Fixed Rate" is the best alternative based on true rate analysis through DEC2002.

Figure 13.9. Loan Comparison Report as of December 2002

The loan comparison report through December 1997 picks the adjustable rate loan as the best alternative, whereas the report through December 2002 shows the fixed rate loan as the better alternative. This implies that if you intend to keep the loan for 10 years or longer, the best alternative is the fixed rate alternative. Otherwise, the adjustable rate loan is the better alternative in spite of the worst-case scenario. Further analysis shows that the actual breakeven of true interest rate occurs at August 2002. That is, the desirable alternative switches from the adjustable rate loan to the fixed rate loan in August 2002.

Note that, under the assumption of worst-case scenario for the rate adjustments, the periodic payment for the adjustable rate loan already exceeds that of the fixed rate loan on December 1997 (as of the rate adjustment on January 1997 to be exact). If the objective were to minimize the periodic payment, the fixed rate loan would have been more desirable as of December 1997. However, all of the other criteria at that point still favor the adjustable rate loan.

Syntax

The following statements are used with PROC LOAN:

PROC LOAN *options* ;
FIXED *options* ;
BALLOON *options* ;
ARM *options* ;
BUYDOWN *options* ;
COMPARE *options* ;

Functional Summary

The statements and options controlling the LOAN procedure are summarized in the following table:

Description	Statement	Option
Statements		
specify an adjustable rate loan	ARM	
specify a balloon payment loan	BALLOON	
specify a buydown rate loan	BUYDOWN	
specify loan comparisons	COMPARE	
specify a fixed rate loan	FIXED	
Data Set Options		
specify output data set for loan summary	PROC LOAN	OUTSUM=
specify output data set for repayment schedule	FIXED	OUT=
specify output data set for loan comparison	COMPARE	OUTCOMP=
Printing Control Options		
suppress printing of loan summary report	FIXED	NOSUMMARYPRINT
suppress all printed output	FIXED	NOPRINT
print amortization schedule	FIXED	SCHEDULE=
suppress printing of loan comparison report	COMPARE	NOCOMPRT
Required Specifications		
specify the loan amount	FIXED	AMOUNT=
specify life of loan as number of payments	FIXED	LIFE=
specify the periodic payment	FIXED	PAYMENT=
specify the initial annual nominal interest rate	FIXED	RATE=
Loan Specifications Options		
specify loan amount as percentage of price	FIXED	AMOUNTPCT=

Description	Statement	Option
specify time interval between compoundings	FIXED	COMPOUND=
specify down payment at loan initialization	FIXED	DOWNPAYMENT=
specify down payment as percentage of price	FIXED	DOWNPAYPCT=
specify amount paid for loan initialization	FIXED	INITIAL=
specify initialization costs as a percent	FIXED	INITIALPCT=
specify time interval between payments	FIXED	INTERVAL=
specify label for the loan	FIXED	LABEL=
specify amount paid for discount points	FIXED	POINTS=
specify discount points as a percent	FIXED	POINTPCT=
specify uniform or lump sum prepayments	FIXED	PREPAYMENTS=
specify the purchase price	FIXED	PRICE=
specify number of decimal places for rounding	FIXED	ROUND=
specify the date of loan initialization	FIXED	START=
Balloon Payment Loan Specification		
Option		
specify the list of balloon payments	BALLOON	BALLOONPAYMENT=
Rate Adjustment Terms Options		
specify frequency of rate adjustments	ARM	ADJUSTFREQ=
specify periodic and life cap on rate adjustment	ARM	CAPS=
specify maximum rate adjustment	ARM	MAXADJUST=
specify maximum annual nominal interest rate	ARM	MAXRATE=
specify minimum annual nominal interest rate	ARM	MINRATE=
Rate Adjustment Case Options		
specify best-case (optimistic) scenario	ARM	BESTCASE
specify predicted interest rates	ARM	ESTIMATEDCASE=
specify constant rate	ARM	FIXEDCASE
specify worst case (pessimistic) scenario	ARM	WORSTCASE
Buydown Rate Loan Specification		
Option		
specify list of nominal interest rates	BUYDOWN	BUYDOWNRATES=
Loan Comparison Options		
specify all comparison criteria	COMPARE	ALL
specify the loan comparison periods	COMPARE	AT=
specify breakeven analysis of the interest paid	COMPARE	BREAKINTEREST
specify breakeven analysis of periodic payment	COMPARE	BREAKPAYMENT
specify minimum attractive rate of return	COMPARE	MARR=
specify present worth of cost analysis	COMPARE	PWOF COST

Description	Statement	Option
specify the income tax rate	COMPARE	TAXRATE=
specify true interest rate analysis	COMPARE	TRUEINTEREST

PROC LOAN Statement

PROC LOAN options ;

The following output option can be used in the PROC LOAN statement. In addition, the following FIXED statement options can be specified in the PROC LOAN statement to be used as defaults for all loans unless otherwise specified for a given loan:

AMOUNT=	INTERVAL=	POINTPCT=
AMOUNTPCT	LABEL=	PREPAYMENTS=
COMPOUND=	LIFE=	PRICE=
DOWNPAYMENT=	NOSUMMARYPRINT	RATE=
DOWNPAYPCT=	NOPRINT	ROUND=
INITIAL=	PAYMENT=	START=
INITIALPCT=	POINTS=	SCHEDULE

(all FIXED statement options other than the OUT= and LABEL= options, which are specific to individual loans).

Output Option

OUTSUM=SAS-data-set

creates an output data set containing loan summary information for all loans other than those for which a different OUTSUM= output data set is specified.

FIXED Statement

FIXED options ;

The FIXED statement specifies a fixed rate and periodic payment loan. It can be specified using the options that are common to all loan statements. The FIXED statement options are listed in this section.

You must specify three of the following options in each loan statement: AMOUNT=, LIFE=, RATE=, and PAYMENT=. The LOAN procedure calculates the fourth parameter based on the values you give the other three. If you specify all four of the options, the PAYMENT= specification is ignored, and the periodic payment is recal-

Part 2. General Information

culated for consistency.

As an alternative to specifying the AMOUNT= option, you can specify the PRICE= option along with one of the following options to facilitate the calculation of the loan amount: AMOUNTPCT=, DOWNPAYMENT=, or DOWNPAYPCT=.

Required Specifications

AMOUNT=*amount*

specifies the loan amount (the outstanding principal balance at the initialization of the loan).

The AMOUNT= option can be abbreviated A=.

LIFE=*n*

gives the life of the loan in number of payments. (The payment frequency is specified by the INTERVAL= option.) For example, if the life of the loan is 10 years with monthly payments, use LIFE=120 and INTERVAL=MONTH (default) to indicate a 10-year loan in which 120 monthly payments are made.

The LIFE= option can be abbreviated L=.

PAYMENT=*amount*

specifies the periodic payment. For ARM and BUYDOWN loans where the periodic payment might change, the PAYMENT= option specifies the initial amount of the periodic payment.

The PAYMENT= option can be abbreviated P=.

RATE=*rate*

specifies the initial annual (nominal) interest rate in percent notation. The rate specified must be in the range 0% to 120%. For example, use RATE=12.75 for a 12.75% loan. For ARM and BUYDOWN loans, where the rate might change over the life of the loan, the RATE= option specifies the initial annual interest rate.

The RATE= option can be abbreviated R=.

Specification Options

AMOUNTPCT=*value*

specifies the loan amount as a percentage of the purchase price (PRICE= option). The AMOUNTPCT= specification is used to calculate the loan amount if the AMOUNT= option is not specified. The value specified must be in the range 1% to 100%.

If both the AMOUNTPCT= and DOWNPAYPCT= options are specified and the sum of their values is not equal to 100, the value of the downpayment percentage is set equal to 100 minus the value of the amount percentage.

The AMOUNTPCT= option can be abbreviated APCT=.

COMPOUND=*time-unit*

specifies the time interval between compoundings. The default is the time unit given by the INTERVAL= option. If the INTERVAL= option is not used, then the default is COMPOUND=MONTH. The following time units are valid COMPOUND= values: CONTINUOUS, DAY, SEMIMONTH, MONTH, QUARTER, SEMIYEAR, and YEAR. The compounding interval is used to calculate the simple interest rate per

payment period from the nominal annual interest rate or vice versa.

DOWNPAYMENT=amount

specifies the down payment at the initialization of the loan. The down payment is included in the calculation of the present worth of cost but not in the calculation of the true interest rate. The after-tax analysis assumes that the down payment is not tax-deductible. (Specify after-tax analysis with the TAXRATE= option in the COMPARE statement.)

The DOWNPAYMENT= option can be abbreviated DP=.

DOWNPAYPCT=value

specifies the down payment as a percentage of the purchase price (PRICE= option). The DOWNPAYPCT= specification is used to calculate the down payment amount if you do not specify the DOWNPAYMENT= option. The value you specify must be in the range 0% to 99%.

If you specified both the AMOUNTPCT= and DOWNPAYPCT= options, and the sum of their values is not equal to 100, the value of the downpayment percentage is set equal to 100 minus the value of the amount percentage.

The DOWNPAYPCT= option can be abbreviated DPCT=.

INITIAL=amount

specifies the amount paid for loan initialization other than the discount points and down payment. This amount is included in the calculation of the present worth of cost and the true interest rate. The after-tax analysis assumes that the initial amount is not tax-deductible. (After-tax analysis is specified by the TAXRATE= option in the COMPARE statement.)

The INITIAL= option can be abbreviated INIT=.

INITIALPCT=value

specifies the initialization costs as a percentage of the loan amount (AMOUNT= option). The INITIALPCT= specification is used to calculate the amount paid for loan initialization if you do not specify the INITIAL= option. The value you specify must be in the range of 0% to 100%.

The INITIALPCT= option can be abbreviated INITPCT=.

INTERVAL=time-unit

gives the time interval between periodic payments. The default is INTERVAL=MONTH. The following time units are valid INTERVAL values: SEMI-MONTH, MONTH, QUARTER, SEMIYEAR, and YEAR.

LABEL='loan-label'

specifies a label for the loan. If you specify the LABEL= option, all output related to the loan is labeled accordingly. If you do not specify the LABEL= option, the loan is labeled by sequence number.

POINTS=amount

specifies the amount paid for discount points at the initialization of the loan. This amount is included in the calculation of the present worth of cost and true interest

Part 2. General Information

rate. The amount paid for discount points is assumed to be tax-deductible in after-tax analysis (that is, if the TAXRATE= option is specified in the COMPARE statement).

The POINTS= option can be abbreviated PNT=.

POINTPCT=value

specifies the discount points as a percentage of the loan amount (AMOUNT= option). The POINTPCT= specification is used to calculate the amount paid for discount points if you do not specify the POINTS= option. The value you specify must be in the range of 0% to 100%.

The POINTPCT= option can be abbreviated PNTPCT=.

PREPAYMENTS=amount

PREPAYMENTS=(date1=prepayment1 date2=prepayment2 ...)

PREPAYMENTS=(period1=prepayment1 period2=prepayment2 ...)

specifies either a uniform prepayment p throughout the life of the loan or lump sum prepayments. A uniform prepayment, p , is assumed to be paid with each periodic payment. Specify lump sum prepayments by pairs of periods (or dates) and respective prepayment amounts.

You can specify the prepayment periods as dates if you specify the START= option. Prepayment periods or dates and the respective prepayment amounts must be in time sequence. The prepayments are treated as principal payments, and the outstanding principal balance is adjusted accordingly. In the adjustable rate and buydown rate loans, if there is a rate adjustment after prepayments, the adjusted periodic payment is calculated based on the outstanding principal balance. The prepayments do not result in periodic payment amount adjustments in fixed rate and balloon payment loans.

The PREPAYMENTS= option can be abbreviated PREP=.

PRICE=amount

specifies the purchase price, which is the loan amount plus the down payment. If you specify the PRICE= option along with the loan amount (AMOUNT= option) or the down payment (DOWNPAYMENT= option), the value of the other one is calculated.

If you specify the PRICE= option with the AMOUNTPCT= or DOWNPAYPCT= options, the loan amount and the downpayment are calculated.

The PRICE= option can be abbreviated PRC=.

ROUND=n

ROUND=NONE

specifies the number of decimal places to which the monetary amounts are rounded for the loan. Valid values for n are integers from 0 to 6. If you specify ROUND=NONE, the values are not rounded off internally, but the printed output is rounded off to two decimal places. The default is ROUND=2.

START=SAS-date

START=yyyy:per

gives the date of loan initialization. The first payment is assumed to be one payment interval after the start date. For example, you can specify the START= option

as '1APR1990'd or as 1990:3 where 3 is the third payment interval. If INTERVAL=QUARTER, 3 refers to the third quarter. If you specify the START= option, all output involving the particular loan is dated accordingly.

The START= option can be abbreviated S=.

Output Options

NOSUMMARYPRINT

suppresses the printing of the loan summary report. The NOSUMMARYPRINT option is usually used when an OUTSUM= data set is created to store loan summary information.

The NOSUMMARYPRINT option can be abbreviated NOSUMPR.

NOPRINT

suppresses all printed output for the loan.

The NOPRINT option can be abbreviated NOP.

OUT=SAS-data-set

writes the loan amortization schedule to an output data set.

OUTSUM=SAS-data-set

writes the loan summary for the individual loan to an output data set.

SCHEDULE

SCHEDULE=*nyears*

SCHEDULE=YEARLY

prints the amortization schedule for the loan. SCHEDULE=*nyears* specifies the number of years the printed amortization table covers. If you omit the number of years or specify a period longer than the loan life, the schedule is printed for the full term of the loan. SCHEDULE=YEARLY prints yearly summary information in the amortization schedule rather than the full amortization schedule. SCHEDULE=YEARLY is useful for long-term loans.

The SCHEDULE option can be abbreviated SCHED.

BALLOON Statement

BALLOON *options;*

The BALLOON statement specifies a fixed rate loan with scheduled balloon payments in addition to the periodic payment. In addition to the required specifications and options listed under the FIXED statement, the following option is used in the BALLOON statement:

BALLOONPAYMENT=(*date1=payment1 date2=payment2 ...*)

BALLOONPAYMENT=(*period1=payment1 period2=payment2 ...*)

specifies pairs of periods and amounts of balloon (lump sum) payments in excess of the periodic payment during the life of the loan. You can also specify the balloon periods as dates if you specify the START= option.

Part 2. General Information

If you do not specify this option, the calculations are identical to a loan specified in a FIXED statement. Balloon periods (or dates) and the respective balloon payments must be in time sequence.

The BALLOONPAYMENT= option can be abbreviated BPAY=.

ARM Statement

ARM options;

The ARM statement specifies an adjustable rate loan where the future interest rates are not known with certainty but will vary within specified limits according to the terms stated in the loan agreement. In practice, the adjustment terms vary. Adjustments in the interest rate can be captured using the ARM statement options.

In addition to the required specifications and options listed under the FIXED statement, you can use the following options with the ARM statement:

Rate Adjustment Terms Options

ADJUSTFREQ=*n*

specifies the number of periods (in terms of the INTERVAL= specification) between rate adjustments. INTERVAL=MONTH ADJUSTFREQ=6 indicates that the nominal interest rate can be adjusted every six months until the life cap or maximum rate (whichever is specified) is reached. The default is ADJUSTFREQ=12. The periodic payment is adjusted every adjustment period even if there is no rate change; therefore, if prepayments are made (as specified with the PREPAYMENTS= option), the periodic payment might change even if the nominal rate does not.

The ADJUSTFREQ= option can be abbreviated ADF=.

CAPS=(*periodic-cap, life-cap*)

specifies the maximum interest rate adjustment (in percent notation) allowed by the loan agreement. The *periodic cap* specifies the maximum adjustment allowed at each adjustment period. The *life cap* specifies the maximum total adjustment over the life of the loan. For example, a loan specified with CAPS=(0.5, 2) indicates that the nominal interest rate can change by 0.5% each adjustment period, and the annual nominal interest rate throughout the life of the loan will be within a 2% range of the initial annual nominal rate.

MAXADJUST=*rate*

specifies the maximum rate adjustment (in percent notation) allowed at each adjustment period. Use the MAXADJUST= option with the MAXRATE= and MINRATE= options. The initial nominal rate plus the maximum adjustment should not exceed the specified MAXRATE= value. The initial nominal rate minus the maximum adjustment should not be less than the specified MINRATE= value.

The MAXADJUST= option can be abbreviated MAXAD=.

MAXRATE=*rate*

specifies the maximum annual nominal rate (in percent notation) that might be

charged on the loan. The maximum annual nominal rate should be greater than or equal to the initial annual nominal rate (the RATE= option value, if specified).

The MAXRATE= option can be abbreviated MAXR=.

MINRATE=rate

specifies the minimum annual nominal rate (in percent notation) that might be charged on the loan. The minimum annual nominal rate should be less than or equal to the initial annual nominal rate (the RATE= option value, if specified).

The MINRATE= option can be abbreviated MINR=.

Rate Adjustment Case Options

PROC LOAN supports four rate adjustment scenarios for analysis of adjustable rate loans: pessimistic (WORSTCASE), optimistic (BESTCASE), no-change (FIXEDCASE), and estimated (ESTIMATEDCASE). The estimated case enables you to analyze the adjustable rate loan using your predictions of future interest rates. The default is worst-case analysis. If more than one case is specified, worst-case analysis is performed. You can specify options for adjustable rate loans as follows:

BESTCASE

specifies a best-case analysis. The best-case analysis assumes the interest rate charged on the loan will reach its minimum allowed limits at each adjustment period and over the life of the loan. If you use the BESTCASE option, you must specify either the CAPS= option or the MINRATE= and MAXADJUST= options.

The BESTCASE option can be abbreviated B.

ESTIMATEDCASE=(date1=rate1 date2=rate2 ...)

ESTIMATEDCASE=(period1=rate1 period2=rate2 ...)

specifies an estimated case analysis that indicates the rate adjustments will follow the rates you predict. This option specifies pairs of periods and estimated nominal interest rates.

The ESTIMATEDCASE= option can be abbreviated ESTC=.

The ESTIMATEDCASE= option can specify adjustments that cannot fit into the BESTCASE, WORSTCASE, or FIXEDCASE specifications, or "what-if" type analysis. If you specify the START= option, you can also specify the estimation periods as dates. Estimated rates and the respective periods must be in time sequence.

If the estimated period falls between two adjustment periods (determined by ADJUSTFREQ= option), the rate is adjusted in the next adjustment period. The nominal interest rate charged on the loan is constant between two adjustment periods.

If any of the MAXRATE=, MINRATE=, CAPS=, and MAXADJUST= options are specified to indicate the rate adjustment terms of the loan agreement, these specifications are used to bound the rate adjustments. By using the ESTIMATEDCASE= option, you are predicting what the annual nominal rates in the market will be at different points in time, not necessarily the interest rate on your particular loan. For example, if the initial nominal rate (RATE= option) is 10, ADJUSTFREQ=6, MAXADJUST=0.5, and the ESTIMATEDCASE=(6=10.5, 12=10.8), the actual nominal rates charged on the loan would be 10.0% initially, 10.5% for the sixth through the

eleventh periods, and 10.8% for the twelfth period onward.

FIXEDCASE

specifies a fixed case analysis that assumes the rate will stay constant. The FIXEDCASE option calculates the ARM loan values similar to a fixed rate loan, but the payments are updated every adjustment period even if the rate does not change, leading to minor differences between the two methods. One such difference is in the way prepayments are handled. In a fixed rate loan, the rate and the payments are never adjusted; therefore, the payment stays the same over the life of the loan even when prepayments are made (instead, the life of the loan is shortened). In an ARM loan with the FIXEDCASE option, on the other hand, if prepayments are made, the payment is adjusted in the following adjustment period (leaving the life of the loan constant).

The FIXEDCASE option can be abbreviated FIXCASE.

WORSTCASE

specifies a worst-case analysis. The worst-case analysis assumes the interest rate charged on the loan will reach its maximum allowed limits at each rate adjustment period and over the life of the loan. If the WORSTCASE option is used, either the CAPS= option or the MAXRATE= and MAXADJUST= options must be specified. The WORSTCASE option can be abbreviated W.

BUYDOWN Statement

BUYDOWN options;

The BUYDOWN statement specifies a buydown rate loan. The buydown rate loans are similar to ARM loans, but the interest rate adjustments are predetermined at the initialization of the loan (usually by paying interest points at the time of loan initialization).

You can use all the required specifications and options listed under the FIXED statement with the BUYDOWN statement. The following option is specific to the BUYDOWN statement and is required:

BUYDOWNRATES=(date1=rate1 date2=rate2 ...)

BUYDOWNRATES=(period1=rate1 period2=rate2 ...)

specifies pairs of periods and the predetermined nominal interest rates that will be charged on the loan starting at the corresponding time periods.

You can also specify the buydown periods as dates if you specify the START=option. Buydown periods (or dates) and the respective buydown rates must be in time sequence.

The BUYDOWNRATES= option can be abbreviated BDR=.

COMPARE Statement

COMPARE options ;

The COMPARE statement compares multiple loans and it can be used with a single loan. You can use only one COMPARE statement. COMPARE statement options specify the periods and desired types of analysis for loan comparison. The default analysis reports the outstanding principal balance, breakeven of payment, breakeven of interest paid, and before-tax true interest rate. The default comparison period corresponds to the first LIFE= option specification. If the LIFE= option is not specified for any loan, the loan comparison period defaults to the first calculated life.

You can use the following options with the COMPARE statement. For more detailed information on loan comparison, see the section "Loan Comparison Details" later in this chapter.

Analysis Options

ALL

is equivalent to specifying the BREAKINTEREST, BREAKPAYMENT, PWOF-COST, and TRUEINTEREST options. The loan comparison report includes all the criteria. You need to specify the MARR= option for present worth of cost calculation.

AT=(date1 date2 ...)

AT=(period1 period2 ...)

specifies the periods for loan comparison reports. If you specify the START= option in the PROC LOAN statement, you can specify the AT= option as a list of dates instead of periods. The comparison periods do not need to be in time sequence. If you do not specify the AT= option, the comparison period defaults to the first LIFE= option specification. If you do not specify the LIFE= option for any of the loans, the loan comparison period defaults to the first calculated life.

BREAKINTEREST

specifies breakeven analysis of the interest paid. The loan comparison report includes the interest paid for each loan through the specified comparison period (AT= option).

The BREAKINTEREST option can be abbreviated BI.

BREAKPAYMENT

specifies breakeven analysis of payment. The periodic payment for each loan is reported for every comparison period specified in the AT=option.

The BREAKPAYMENT option can be abbreviated BP.

MARR=rate

specifies the MARR (Minimum Attractive Rate of Return) in percent notation. MARR reflects the cost of capital or the opportunity cost of money. The MARR= option is used in calculating the present worth of cost.

PWOF-COST

calculates the present worth of cost (net present value of costs) for each loan based

Part 2. General Information

on the cash flow through the specified comparison periods. The calculations account for down payment, initialization costs and discount points, as well as the payments and outstanding principal balance at the comparison period. If you specify the `TAXRATE=` option, the present worth of cost is based on after-tax cash flow. Otherwise, before-tax present worth of cost is calculated. You need to specify the `MARR=` option for present worth of cost calculations.

The `PWOF COST` option can be abbreviated `PWC`.

TAXRATE=rate

specifies income tax rate in percent notation for the after-tax calculations of the true interest rate and present worth of cost for those assets that qualify for tax deduction. If you specify this option, the amount specified in the `POINTS=` option and the interest paid on the loan are assumed to be tax-deductible. Otherwise, it is assumed that the asset does not qualify for tax deductions, and the cash flow is not adjusted for tax savings.

The `TAXRATE=` option can be abbreviated `TAX=`.

TRUEINTEREST

calculates the true interest rate (effective interest rate based on the cash flow of all payments, initialization costs, discount points, and the outstanding principal balance at the comparison period) for all the specified loans through each comparison period. If you specify the `TAXRATE=` option, the true interest rate is based on after-tax cash flow. Otherwise, the before-tax true interest rate is calculated.

The `TRUEINTEREST` option can be abbreviated `TI`.

Output Options

NOCOMPRINT

suppresses the printing of the loan comparison report. The `NOCOMPRINT` option is usually used when an `OUTCOMP=` data set is created to store loan comparison information.

The `NOCOMPRINT` option can be abbreviated `NOCP`.

OUTCOMP=SAS-data-set

writes the loan comparison report to an output data set.

Details

Computational Details

These terms are used in the formulas that follow:

p	periodic payment
a	principal amount
r_a	nominal annual rate
f	compounding frequency (per year)
f'	payment frequency (per year)
r	periodic rate
r_e	effective interest rate
n	total number of payments

The periodic rate, or the simple interest applied during a payment period, is given by

$$r = \left(1 + \frac{r_a}{f}\right)^{f/f'} - 1$$

Note that the interest calculation is performed at each payment period rather than at the compound period. This is done by adjusting the nominal rate. Refer to Muksian (1984) for details.

Note that when $f = f'$, that is, when the payment and compounding frequency coincide, the preceding expression reduces to the familiar form:

$$r = \frac{r_a}{f}$$

The periodic rate for continuous compounding can be obtained from this general expression by taking the limit as the compounding frequency f goes to infinity. The resulting expression is

$$r = \exp\left(\frac{r_a}{f'}\right) - 1$$

The effective interest rate, or annualized percentage rate (APR), is that rate which, if compounded once per year, is equivalent to the nominal annual rate compounded f times per year. Thus,

$$(1 + r_e) = (1 + r)^f = \left(1 + \frac{r_a}{f}\right)^f$$

Part 2. General Information

or

$$r_e = \left(1 + \frac{r_a}{f}\right)^f - 1$$

For continuous compounding, the effective interest rate is given by

$$r_e = \exp(r_a) - 1$$

Refer to Muksian (1984) for details.

The payment is calculated as

$$p = \frac{ar}{1 - \frac{1}{(1+r)^n}}$$

The amount is calculated as

$$a = \frac{p}{r} \left(1 - \frac{1}{(1+r)^n}\right)$$

Both the payment and amount are rounded to the nearest hundredth (cent) unless the ROUND= specification is different than the default, 2.

The total number of payments n is calculated as

$$n = \frac{-\ln\left(1 - \frac{ar}{p}\right)}{\ln(1+r)}$$

The total number of payments is rounded up to the nearest integer.

The nominal annual rate is calculated using the bisection method, with a as the objective and r starting in the interval between $8 * 10^{-6}$ and .1 with an initial midpoint .01 and successive midpoints bisecting.

Loan Comparison Details

In order to compare the costs of different alternatives, the input cash flow for the alternatives must be represented in equivalent values. The equivalent value of a cash flow accounts for the time-value of money. That is, it is preferable to pay the same amount of money later than to pay it now, since the money can earn interest while you keep it. The MARR (**M**inimum **A**tractive **R**ate of **R**eturn) reflects the cost of capital or the opportunity cost of money, that is, the interest that would have been earned on the savings that is foregone by making the investment. The MARR is used to discount the cash flow of alternatives into equivalent values at a fixed point in time. The MARR can vary for each investor and for each investment. Therefore,

the MARR= option must be specified in the COMPARE statement if present worth of cost (PWOF COST option) comparison is specified.

Present worth of cost reflects the equivalent amount at loan initialization of the loan cash flow discounted at MARR, not accounting for inflation. Present worth of cost accounts for the down payment, initialization costs, discount points, periodic payments, and the principal balance at the end of the report period. Therefore, it reflects the present worth of cost of the asset, not the loan. It is only meaningful to use minimization of present worth of cost as a selection criterion if the assets (down payment plus loan amount) are of the same value.

Another economic selection criterion is the rate of return (internal rate of return) of the alternatives. If interest is being earned by an alternative, the objective would be to maximize the rate of return. If interest is being paid, as in loan alternatives, the best alternative is the one that minimizes the rate of return. The true interest rate reflects the effective annual rate charged on the loan based on the cash flow, including the initialization cost and the discount points.

The effects of taxes on different alternatives must be accounted for when these vary among different alternatives. Since interest costs on certain loans are tax-deductible, the comparisons for those loans are made based on the after-tax cash flows. The cost of the loan is reduced by the tax benefits it offers through the loan life if the TAXRATE= option is specified. The present worth of cost and true interest rate are calculated based on the after-tax cash flow of the loan. The down payment on the loan and initialization costs are assumed to be not tax-deductible in after-tax analysis. Discount points and the interest paid in each periodic payment are assumed to be tax-deductible if the TAXRATE= option is specified. If the TAXRATE= option is not specified, the present worth of cost and the true interest rate are based on before-tax cash flow, assuming that the interest paid on the specified loan does not qualify for tax benefits.

The other two selection criteria are breakeven analysis of periodic payment and interest paid. If the objective is to minimize the periodic payment, the best alternative would be the one with the minimum periodic payment. If the objective is to minimize the interest paid on the principal, then the best alternative is the one with the least interest paid.

Another criterion might be the minimization of the outstanding balance of the loan at a particular point in time. For example, if you plan to sell a house before the end of the loan life (which is often the case), you might want to select the loan with the minimum principal balance at the time of the sale, since this balance must be paid at that time. The outstanding balance of the alternative loans is calculated for each loan comparison period by default.

If you specified the START= option in the PROC LOAN statement, the present worth of cost reflects the equivalent amount for each loan at that point in time. Any loan that has a START= specification different from the one in the PROC LOAN statement is not processed in the loan comparison.

The loan comparison report for each comparison period contains for each loan the loan label, outstanding balance, and any of the following measures if requested in the

COMPARE statement: periodic payment (BREAKPAY option), total interest paid to date (BREAKINTEREST option), present worth of cost (PWOFCOST option), and true interest rate (TRUEINTEREST option). The best loan is selected on the basis of present worth of cost or true interest rate. If both PWOFCOST and TRUEINTEREST options are specified, present worth of cost is the basis for the selection of the best loan.

You can use the OUTCOMP= option in the COMPARE statement to write the loan comparison report to a data set. The NOCOMPRINT option suppresses the printing of a loan comparison report.

OUT= Data Set

The OUT= option writes the loan amortization schedule to an output data set. The OUT= data set contains one observation for each payment period (or one observation for each year if you specified the SCHEDULE=YEARLY option). If you specified the START= option, the DATE variable denotes the date of the payment. Otherwise, YEAR and period variable (SEMIMONTH, MONTH, QUARTER, or SEMIYEAR) denote the payment year and period within the year.

The OUT= data set contains the following variables:

- DATE, date of the payment. DATE is included in the OUT= data set only when you specify the START= option.
- YEAR, year of the payment period. YEAR is included in the OUT= data set only when you do not specify the START= option.
- PERIOD, period within the year of the payment period. The name of the period variable matches the INTERVAL= specification (SEMIMONTH, MONTH, QUARTER, or SEMIYEAR.) The PERIOD variable is included in the OUT= data set only when you do not specify the START= option.
- BEGPRIN, beginning principal balance
- PAYMENT, payment
- INTEREST, interest payment
- PRIN, principal repayment
- ENDPRIN, ending principal balance

OUTCOMP= Data Set

The OUTCOMP= option in the COMPARE statement writes the loan comparison analysis results to an output data set. If you specified the START= option, the DATE variable identifies the date of the loan comparison. Otherwise, the PERIOD variable identifies the comparison period.

The OUTCOMP= data set contains one observation for each loan for each loan comparison period. The OUTCOMP= data set contains the following variables:

- DATE, date of loan comparison report. The DATE variable is included in the OUTCOMP= data set only when you specify the START= option.
- PERIOD, period of the loan comparison for the observation. The PERIOD variable is included in the OUTCOMP= data set only when you do not specify the START= option.
- LABEL, label string for the loan
- TYPE, type of the loan
- PAYMENT, periodic payment at the time of report. The PAYMENT is included in the OUTCOMP= data set if you specified the BREAKPAYMENT or ALL option or if you used default criteria.
- INTPAY, interest paid through the time of report. The INTPAY variable is included in the OUTCOMP= data set if you specified the BREAKINTEREST or ALL option or if you used default criteria.
- TRUERATE, true interest rate charged on the loan. The TRUERATE variable is included in the OUTCOMP= data set if you specified the TRUERATE or ALL option or if you used default criteria.
- PWOF COST, present worth of cost. The PWOF COST variable is included in the OUTCOMP= data set only if you specified the PWOF COST or ALL option.
- BALANCE, outstanding principal balance at the time of report

OUTSUM= Data Set

The OUTSUM= option writes the loan summary to an output data set. If you specified this option in the PROC LOAN statement, the loan summary information for all loans will be written to the specified data set, except for those loans for which you specified a different OUTSUM= data set on the ARM, BALLOON, BUYDOWN, or FIXED statement.

You can also specify the OUTSUM= option in individual loan statements, in which case the loan summary information of the individual loan is written to the specified data set. The OUTSUM= data set contains one observation for each loan and contains the following variables:

- TYPE, type of loan
- LABEL, loan label
- PAYMENT, periodic payment
- AMOUNT, loan principal
- DOWNPAY, down payment. DOWNPAY is included in the OUTSUM= data set only when you specify a down payment.
- INITIAL, loan initialization costs. INITIAL is included in the OUTSUM= data set only when you specify initialization costs.

Part 2. General Information

- POINTS, discount points. POINTS is included in the OUTSUM= data set only when you specify discount points.
- TOTAL, total payment
- INTEREST, total interest paid
- RATE, nominal annual interest rate
- EFFRATE, effective interest rate
- INTERVAL, payment interval
- COMPOUND, compounding interval
- LIFE, loan life (that is, the number of payment intervals)
- NCOMPND, number of compounding intervals
- COMPUTE, computed loan parameter: life, amount, payment, or rate

If you specified the START= option either in the PROC LOAN statement or for the individual loan, the OUTSUM= data set also contains the following variables:

- BEGIN, start date
- END, loan termination date

Printed Output

The output from PROC LOAN consists of the loan summary table, loan amortization schedule, and loan comparison report.

Loan Summary Table

The loan summary table shows the total payment and interest, the initial nominal annual and effective interest rates, payment and compounding intervals, the length of the loan in the time units specified, the start and end dates (if specified), a list of nominal and effective interest rates, and periodic payments throughout the life of the loan.

A list of balloon payments for balloon payment loans and a list of prepayments (if specified) are printed with their respective periods (or dates).

The loan summary table is printed for each loan by default. The NOSUMMARYPRINT option specified in the PROC LOAN statement will suppress the printing of the loan summary table for all loans. The NOSUMMARYPRINT option can be specified in individual loan statements to selectively suppress the printing of the loan summary table.

Loan Repayment Schedule

The amortization schedule contains for each payment period the year and period within the year (or date, if you specified the START= option), principal balance at the beginning of the period, total payment, interest payment and principal payment for the period, and the principal balance at the end of the period. If you specified the

SCHEDULE=YEARLY option, the amortization will contain a summary for each year instead of for each payment period.

The amortization schedule is not printed by default. The SCHEDULE option in the PROC LOAN statement requests the printing of amortization tables for all loans. You can specify the SCHEDULE option in individual loan statements to selectively request the printing of the amortization schedule.

Loan Comparison Report

The loan comparison report is processed for each report period and contains the results of economic analysis of the loans. The quantities reported can include the outstanding principal balance, after-tax or before-tax present worth of cost and true interest rate, periodic payment, and the interest paid through the report period for each loan. The best alternative is identified if the asset value (down payment plus loan amount) is the same for each alternative.

The loan comparison report is printed by default. The NOCOMPRINT option specified in the COMPARE statement suppresses the printing of the loan comparison report.

ODS Table Names

PROC LOAN assigns a name to each table it creates. You can use these names to reference the table when using the Output Delivery System (ODS) to select tables and create output data sets. These names are listed in the following table. For more information on ODS, see Chapter 6, “Using the Output Delivery System.”

Table 13.1. ODS Tables Produced in PROC LOAN

ODS Table Name	Description	Option
ODS Tables Created by the PROC LOAN, FIXED, ARM, BALLOON, and BUYDOWN Statements		
Repayment	Loan Repayment Schedule	SCHEDULE
ODS Tables Created by the FIXED, ARM, BALLOON, and BUYDOWN Statements		
LoanSummary	Loan Summary	default
RateList	Rates and Payments	default
PrepayList	Prepayments and Periods	PREPAYMENTS=
ODS Tables Created by the BALLOON Statement		
BalloonList	Balloon Payments and Periods	default
ODS Tables Created by the COMPARE Statement		
Comparison	Loan Comparison Report	

Part 2. General Information

Table 13.1. (continued)

ODS Table Name	Description	Option
----------------	-------------	--------

Examples

Example 13.1. Discount Points for Lower Interest Rates

This example illustrates the comparison of two \$100,000 loans. The major difference between the two loans is that the nominal interest rate in the second one is lower than the first with the added expense of paying discount points at the time of initialization.

Both alternatives are 30-year loans. The first loan is labeled "8.25% - no discount points" and the second one is labeled "8% - 1 discount point."

Assume that the interest paid qualifies for a tax deduction, and you are in the 33% tax bracket. Also, your Minimum Attractive Rate of Return (MARR) for an alternative investment is 4% (adjusted for tax rate.)

You use the following statements to find the breakeven point in the life of the loan for your preference between the loans:

```
proc loan start=1992:1 nosummaryprint amount=100000 life=360;
  fixed rate=8.25 label='8.25% - no discount points';
  fixed rate=8 points=1000 label='8% - 1 discount point';
  compare at=(48 54 60) all taxrate=33 marr=4;
run;
```

Output 13.1.1 shows the loan comparison reports as of January 1996 (48th period), July 1996 (54th period), and January 1997(60th period.)

Output 13.1.1. Loan Comparison Reports for Discount Point Breakeven

The LOAN Procedure					
Loan Comparison Report					
Analysis through JAN1996					
Loan Label	Ending Outstanding	Present Worth of Cost	Payment	Interest Paid	True Rate
8.25% - no discount points	96388.09	105546.17	751.27	32449.05	5.67
8% - 1 discount point	96219.32	105604.05	733.76	31439.80	5.69
NOTE: "8.25 - no discount points" is the best alternative based on present worth of cost analysis through JAN1996.					
Loan Comparison Report					
Analysis through JUL1996					
Loan Label	Ending Outstanding	Present Worth of Cost	Payment	Interest Paid	True Rate
8.25% - no discount points	95847.27	106164.97	751.27	36415.85	5.67
8% - 1 discount point	95656.22	106153.97	733.76	35279.26	5.67
NOTE: "8 - 1 discount point" is the best alternative based on present worth of cost analysis through JUL1996.					

Part 2. General Information

The LOAN Procedure					
Loan Comparison Report					
Analysis through JAN1997					
Loan Label	Ending Outstanding	Present Worth of Cost	Payment	Interest Paid	True Rate
8.25% - no discount points	95283.74	106768.07	751.27	40359.94	5.67
8% - 1 discount point	95070.21	106689.80	733.76	39095.81	5.66

NOTE: "8 - 1 discount point" is the best alternative based on present worth of cost analysis through JAN1997.

Notice that the breakeven point for present worth of cost and true rate both happen on July 1996. This indicates that if you intend to keep the loan for 4.5 years or more, it is better to pay the discount points for the lower rate. If your objective is to minimize the interest paid or the periodic payment, the "8% - 1 discount point" loan is the preferred choice.

Example 13.2. Refinancing a Loan

Assume that you obtained a fixed rate 15-year loan in June 1990 for \$78,500 with a nominal annual rate of 10%. By early 1992, the market offers a 7.5% interest rate, and you are considering whether to refinance your loan.

Use the following statements to find out the status of the loan on February 1992. Output 13.2.1 shows the results:

```
proc loan start=1990:6;
  fixed life=180 rate=10 amount=78500 noprint
  label='Original loan';
  compare at=('10FEB92'd) ;
run;
```

Output 13.2.1. Loan Comparison Report for Original Loan

The LOAN Procedure				
Loan Comparison Report				
Analysis through FEB1992				
Loan Label	Ending Outstanding	Payment	Interest Paid	True Rate
Original loan	74396.51	843.57	12767.91	10.47

The monthly payment on the original loan is \$843.57. The ending outstanding principal balance as of February is \$74,396.51. At this point, you might want to refinance your loan with another 15-year loan. The alternate loan has a 7.5% nominal annual rate. The initialization costs are \$1,419.00. Use the following statements to compare your alternatives:

```
proc loan start=1992:2 amount=74396.51;
```

```

fixed rate=10 payment=843.57
      label='Keep the original loan' noprint;
fixed life=180 rate=7.5 init=1419
      label='Refinance at 7.5%' noprint;
compare at=(14 15) taxrate=33 marr=4 all;
run;

```

Output 13.2.2. Loan Comparison Report for Refinancing Decision

The LOAN Procedure					
Loan Comparison Report					
Analysis through APR1993					
Loan Label	Ending Outstanding	Present Worth of Cost	Payment	Interest Paid	True Rate
Keep the original loan	71090.78	76636.65	843.57	8504.25	6.91
Refinance at 7.5%	71119.93	76666.00	689.66	6378.66	6.98
NOTE: "Keep the original loan" is the best alternative based on present worth of cost analysis through APR1993.					
Loan Comparison Report					
Analysis through MAY1993					
Loan Label	Ending Outstanding	Present Worth of Cost	Payment	Interest Paid	True Rate
Keep the original loan	70839.63	76788.82	843.57	9096.67	6.91
Refinance at 7.5%	70874.77	76723.79	689.66	6823.16	6.86
NOTE: "Refinance at 7.5" is the best alternative based on present worth of cost analysis through MAY1993.					

The comparison reports of April 1993 and May 1993 in Output 13.2.2 illustrate the breakeven between the two alternatives. If you intend to keep the loan through May 1993 or longer, your initialization costs for the refinancing are justified. The periodic payment of the refinanced loan is \$689.66.

Example 13.3. Prepayments on a Loan

This example compares a 30-year loan with and without prepayments. Assume the 30-year loan has an 8.25% nominal annual rate. Use the following statements to see the effect of making uniform prepayments of \$500 with periodic payment:

```

proc loan start=1992:12 rate=8.25 amount=240000 life=360;
  fixed label='No prepayments';
  fixed label='With Prepayments' prepay=500 ;
  compare at=(120) taxrate=33 marr=4 all;
run;

```

Part 2. General Information

Output 13.3.1. Loan Summary Reports and Loan Comparison Report

The LOAN Procedure			
Fixed Rate Loan Summary			
No prepayments			
Downpayment	0.00	Principal Amount	240000.00
Initialization	0.00	Points	0.00
Total Interest	409094.17	Nominal Rate	8.25%
Total Payment	649094.17	Effective Rate	8.57%
Pay Interval	MONTHLY	Compounding	MONTHLY
No. of Payments	360	No. of Compoundings	360
Start Date	DEC1992	End Date	DEC2022

Rates and Payments for No prepayments			
Date	Nominal Rate	Effective Rate	Payment
DEC1992	8.25%	8.57%	1803.04

The LOAN Procedure			
Fixed Rate Loan Summary			
With Prepayments			
Downpayment	0.00	Principal Amount	240000.00
Initialization	0.00	Points	0.00
Total Interest	183650.70	Nominal Rate	8.25%
Total Payment	423650.70	Effective Rate	8.57%
Pay Interval	MONTHLY	Compounding	MONTHLY
No. of Payments	184	No. of Compoundings	184
Start Date	DEC1992	End Date	APR2008

Rates and Payments for With Prepayments			
Date	Nominal Rate	Effective Rate	Payment
DEC1992	8.25%	8.57%	2303.04

The LOAN Procedure					
Loan Comparison Report					
Analysis through DEC2002					
Loan Label	Ending Outstanding	Present Worth of Cost	Payment	Interest Paid	True Rate
No prepayments	211608.05	268762.31	1803.04	187972.8	5.67
With Prepayments	118848.23	264149.25	2303.04	155213.0	5.67

NOTE: "With Prepayments" is the best alternative based on present worth of cost analysis through DEC2002.

Notice that with prepayments you pay off the loan in slightly more than 15 years. Also, the total payments and total interest are considerably lower with the prepayments. If you can afford the prepayments of \$500 each month, another alternative you should consider is using a 15-year loan, which is generally offered at a lower nominal interest rate.

Example 13.4. Output Data Sets

This example shows the analysis and comparison of five alternative loans. Initialization cost, discount points, and both lump sum and periodic payments are included in the specification of these loans. Although no printed output is produced, the loan summary and loan comparison information is stored in the OUTSUM= and OUTCOMP= data sets. Output 13.4.1 illustrates the contents of the output data sets.

```
proc loan start=1992:12 noprint outsum=loans
      amount=100000 life=360;

      fixed rate=13 life=180 prepayment=200
      label='BANK1, Fixed Rate';

      arm rate=10.5 estimatedcase=(12=11.5 18=12)
      label='BANK1, Adjustable Rate';

      buydown rate=12 interval=semimonth init=15000
      bdrates=(3=14 10=16) label='BANK2, Buydown';

      arm rate=10.8 worstcase caps=(0.5, 2.5)
      adjustfreq=6 label='BANK3, Adjustable Rate'
      prepayments=(12=2000 36=5000);

      balloon rate=13 life=480
      points=1100 balloonpayment=(15=2000 48=1000)
      label='BANK4, with Balloon Payment';

      compare at=(120 360) all marr=12 tax=33 outcomp=comp;
run;

proc print data=loans;
run;

proc print data=comp;
run;
```

Output 13.4.1. OUTSUM= and OUTCOMP= Data Sets

Obs	TYPE	LABEL	PAYMENT	AMOUNT	INITIAL	POINTS	TO-		
TAL		INTEREST							
1	FIXED	BANK1, Fixed Rate	1465.24	100000	0	0	182845.49	82845.49	
2	ARM	BANK1, Adjustable Rate	914.74	100000	0	0	367829.20	267829.20	
3	BUYDOWN	BANK2, Buydown	599.55	100000	15000	0	262965.77	162965.77	
4	ARM	BANK3, Adjustable Rate	937.24	100000	0	0	382789.77	282789.77	
5	BALLOON	BANK4, with Balloon Payment	1064.48	100000	0	1100	513955.78	413955.78	
Obs	RATE	EFFRATE	INTERVAL	COMPOUND	LIFE	NCOMPND	COMPUTE	START	END
1	0.130	0.13803	MONTHLY	MONTHLY	125	125	PAYMENT	DEC1992	MAY2003
2	0.105	0.11020	MONTHLY	MONTHLY	360	360	PAYMENT	DEC1992	DEC2022
3	0.120	0.12716	SEMIMONTHLY	SEMIMONTHLY	360	360	PAYMENT	DEC1992	DEC2007
4	0.108	0.11351	MONTHLY	MONTHLY	360	360	PAYMENT	DEC1992	DEC2022
5	0.130	0.13803	MONTHLY	MONTHLY	480	480	PAYMENT	DEC1992	DEC2032

Part 2. General Information

Obs	DATE	TYPE	LABEL	PAYMENT	INTEREST	TRUERATE	PWOF COST	BALANCE
1	DEC2002	FIXED	BANK1, Fixed Rate	1465.24	82629.51	0.09066	86219.64	6800.71
2	DEC2002	ARM	BANK1, Adjustable Rate	1026.73	114661.68	0.08143	76281.06	93246.64
3	DEC2002	BUYDOWN	BANK2, Buydown	732.51	135439.05	0.14623	107791.10	60374.65
4	DEC2002	ARM	BANK3, Adjustable Rate	1046.81	119300.98	0.08888	80958.61	87745.00
5	DEC2002	BALLOON	BANK4, with Balloon Payment	1064.48	126966.27	0.09189	82001.89	96228.67
6	DEC2022	FIXED	BANK1, Fixed Rate	1155.73	82845.49	0.09066	86203.45	0.00
7	DEC2022	ARM	BANK1, Adjustable Rate	1026.76	267829.20	0.08209	69197.06	0.00
8	DEC2022	BUYDOWN	BANK2, Buydown	732.68	162965.77	0.14368	107249.24	0.00
9	DEC2022	ARM	BANK3, Adjustable Rate	1046.79	282789.77	0.09011	75684.48	0.00
10	DEC2022	BALLOON	BANK4, with Balloon Payment	1064.48	357507.19	0.09145	75031.89	71294.39

References

- DeGarmo, E.P., Sullivan, W.G., and Canada, J.R. (1984), *Engineering Economy*, Seventh Edition, New York: Macmillan Publishing Company.
- Muksian, R. (1984), *Financial Mathematics Handbook*, Englewood Cliffs: Prentice-Hall.
- Newnan, D.G. (1988), *Engineering Economic Analysis*, Third Edition, San Jose, California: Engineering Press.
- Riggs, J.L. and West, T.M. (1986), *Essentials of Engineering Economics*, Second Edition, New York: McGraw Hill, Inc.